Chapter 5: Manipulating Expressions with Maple V

5.1 Using simplify, side relations, and assume

Try It! (p. 120)

Find the simplest expression that is equivalent to
$\sin(x)^7+\sin(x)^5\cos(x)+\sin(x)^5\cos(x)^2-\sin(x)^3\cos(x)^3$.

Solution

```
> restart;
```

Begin with

```
> EXPR := sin(x)^7  - sin(x)^5*cos(x) + sin(x)^5*cos(x)^2 - sin(x)^3*cos(x)^3;
```

$$EXPR := \sin(x)^7+\sin(x)^5\cos(x)^2-\sin(x)^5\cos(x)-\sin(x)^3\cos(x)^3$$

As a first attempt, try the basic simplification:

```
> simplify( EXPR );
```

$$-\sin(x)\cos(x)+\sin(x)-2\sin(x)\cos(x)^2+\sin(x)\cos(x)^3+\sin(x)\cos(x)^4$$

While this has reduced the order of the highest exponents, this does not appear to be any simpler than the original expression. (Note that the highest powers now occur on the cosine terms.)

Using the same approach as in Example 5-2 (p. 120), simplification with a preference towards sine terms yields:

```
> EXPRs := simplify( EXPR, { sin(x)^2 + cos(x)^2 = 1 }, [ cos(x), sin(x) ] );
```

$$EXPRs := -\sin(x)^3\cos(x)+\sin(x)^5$$

This is much simpler! In fact, factor should be able to make further improvements:

```
> EXPRs := factor( EXPRs );
```

$$EXPRs := \sin(x)^3(-\cos(x)+\sin(x)^2)$$

Another approach to the problem is to first factor the expression. Unless special care is taken to ensure the factored form is not lost, this approach is not likely to be effective. (Try It!)

```
>
```

Try It! (p. 122)

Repeat the simplification of the expression in Example 5-3 for each combination of two assumptions on x, y, and q. Explain your results.

Solution

```
> restart;
> EXPR := (x*y^4)^(3/(q+1)):
> EXPR = simplify( EXPR );
```

$$(x\,y^4)^{\left(\frac{3}{q+1}\right)}=(x^3y^{12})^{\left(\frac{1}{q+1}\right)}$$

```
>
```

Original: all three assumptions

```
> assume( q>-1, y>0, x>0 );
> about( q, x, y );
```

```
Originally q, renamed q~:
  is assumed to be: RealRange(Open(-1),infinity)

Originally x, renamed x~:
  is assumed to be: RealRange(Open(0),infinity)

Originally y, renamed y~:
  is assumed to be: RealRange(Open(0),infinity)
```

```
> EXPR = simplify( EXPR );
```

$$(x\sim y\sim^4)^{\left(\frac{3}{q\sim+1}\right)}=x\sim^{\left(\frac{3}{q\sim+1}\right)}\,y\sim^{\left(\frac{12}{q\sim+1}\right)}$$

```
>
```

Case 1: No assumption on x

```
> x := 'x': y := 'y': q := 'q':
> assume( q>-1, y>0 );
> about( q, x, y );
Originally q, renamed q~:
  is assumed to be: RealRange(Open(-1),infinity)

x:
  nothing known about this object

Originally y, renamed y~:
  is assumed to be: RealRange(Open(0),infinity)
> EXPR = simplify( EXPR );
```

$$(x\,y\sim^4)^{\left(\frac{3}{q\sim+1}\right)} = y\sim^{\left(\frac{12}{q\sim+1}\right)}\,(x^3)^{\left(\frac{1}{q\sim+1}\right)}$$

```
>
```

Case 2: No assumption on y

```
> x := 'x': y := 'y': q := 'q':
> assume( q>-1, x>0 );
> about( q, x, y );
Originally q, renamed q~:
  is assumed to be: RealRange(Open(-1),infinity)

Originally x, renamed x~:
  is assumed to be: RealRange(Open(0),infinity)

y:
  nothing known about this object
> EXPR = simplify( EXPR );
```

$$(x\sim\,y^4)^{\left(\frac{3}{q\sim+1}\right)} = x\sim^{\left(\frac{3}{q\sim+1}\right)}\,(y^{12})^{\left(\frac{1}{q\sim+1}\right)}$$

```
>
```

Case 3: No assumption on q

```
> x := 'x': y := 'y': q := 'q':
> assume( y>0, x>0 );
> about( q, x, y );
q:
  nothing known about this object

Originally x, renamed x~:
  is assumed to be: RealRange(Open(0),infinity)

Originally y, renamed y~:
  is assumed to be: RealRange(Open(0),infinity)
> EXPR = simplify( EXPR );
```

$$(x\sim\,y\sim^4)^{\left(\frac{3}{q+1}\right)} = x\sim^{\left(\frac{3}{q+1}\right)}\,y\sim^{\left(\frac{12}{q+1}\right)}$$

```
>
```

The assumptions on x and y provide the information Maple needs to be sure the simplifications for those terms are appropriate. The assumption on q is not needed.

```
>
```

5.2 Using normal

Try It! (p. 124)

To further understand the different ways in which simplify and normal work, look at -- and explain -- the results of applying simplify and normal to the numerator and denominator of the trigonometric expression in the Example 5-4.

Solution

```
> restart;
```

The expression to be analyzed here is created as in Example 5-4 (p. 123).

```
> EXPR1 := (x^10-1)/(x^2-1):
> EXPR3 := subs( x=sin(theta), EXPR1 );
```

The numerator and denominator are obtained with numer and denom, respectively.

```
> top := numer( EXPR3 );
> bottom := denom( EXPR3 );
```

$$top := \sin(\theta)^{10} - 1$$

$$bottom := \sin(\theta)^2 - 1$$

```
>
```

The application of simplify to the numerator and denominator yields

```
> simplify( top );
> simplify( bottom );
```

$$-5\cos(\theta)^2 + 10\cos(\theta)^4 - 10\cos(\theta)^6 + 5\cos(\theta)^8 - \cos(\theta)^{10}$$

$$-\cos(\theta)^2$$

In these cases Maple has converted all powers of $\sin(\theta)$ into appropriate expressions involving $\cos(\theta)$. Since both numerator and denominator have a common factor of $\cos(\theta)^2$, their ratio is only of degree 8 in $\cos(\theta)$.

```
> simplify( EXPR3 );
```

$$5 - 10\cos(\theta)^2 + 10\cos(\theta)^4 - 5\cos(\theta)^6 + \cos(\theta)^8$$

```
>
```

Since normal is intended for use with rational expressions, we do not expect to any changes when normal is applied separately to the numerator and denominator.

```
> normal( top );
> normal( bottom );
```

$$\sin(\theta)^{10} - 1$$

$$\sin(\theta)^2 - 1$$

However, as discussed in Example 5-4, normal does detect the common factor in the rational expression.

```
> normal( EXPR3 );
```

$$\sin(\theta)^8 + \sin(\theta)^6 + \sin(\theta)^4 + \sin(\theta)^2 + 1$$

In addition to the comments at the end of Example 5-4, the equivalence of the results from simplify and normal can be seen by applying simplify to the normalized expression.

```
> simplify( " );
```

$$5 - 10\cos(\theta)^2 + 10\cos(\theta)^4 - 5\cos(\theta)^6 + \cos(\theta)^8$$

```
>
```

5.3 Using factor

Try It! (p. 126)

Find the factorization of $\frac{x^4 - y^4}{x^3 - y^3}$. Compare the results from factor with those from simplify and normal.

Solution

```
> restart;
```

The rational expression to be studied is

```
> EXPR := (x^4-y^4)/(x^3-y^3);
```

$$EXPR := \frac{x^4 - y^4}{x^3 - y^3}$$

When factor, simplify, and normal are applied to this expression we obtain

```
> factor( EXPR );
> simplify( EXPR );
> normal( EXPR );
```

$$\frac{(y + x)(x^2 + y^2)}{x^2 + x\,y + y^2}$$

$$\frac{x^3 + y\,x^2 + y^2\,x + y^3}{x^2 + x\,y + y^2}$$

$$\frac{x^3 + y\,x^2 + y^2\,x + y^3}{x^2 + x\,y + y^2}$$

While each of the three commands operates differently, the only difference in the results is that the result from factor is factored while those from simplify and normal are expanded.

```
>
```

5.4 Using expand and combine

Try It! (p. 128)

To better understand how combine works, use op to extract the three terms from EXPRe, apply combine to each term, then reassemble the results. Is this the same as EXPR? Explain.

Solution

```
> restart;
```

The expression and its equivalent expanded form are obtained as in Example 5-8 (p. 128).

```
> EXPR := cos( 2*theta + phi );
> EXPRe := expand( EXPR );
```

$$EXPR := \cos(2\,\theta + \phi)$$

$$EXPRe := 2\cos(\phi)\cos(\theta)^2 - \cos(\phi) - 2\sin(\phi)\sin(\theta)\cos(\theta)$$

The three terms in the expanded form of the expression can be isolated as follows:

```
> TERM1 := op( 1, EXPRe );
> TERM2 := op( 2, EXPRe );
> TERM3 := op( 3, EXPRe );
```

$$TERM1 := 2\cos(\phi)\cos(\theta)^2$$

$$TERM2 := -\cos(\phi)$$

$$TERM3 := -2\sin(\phi)\sin(\theta)\cos(\theta)$$

The application of combine to the first and third terms appears to create more complicated expressions

```
> TERM1c := combine( TERM1 );
> TERM2c := combine( TERM2 );
> TERM3c := combine( TERM3 );
```

$$TERM1c := \frac{1}{2}\cos(-2\,\theta + \phi) + \frac{1}{2}\cos(2\,\theta + \phi) + \cos(\phi)$$

$$TERM2c := -\cos(\phi)$$

$$TERM3c := -\frac{1}{2}\cos(-2\,\theta + \phi) + \frac{1}{2}\cos(2\,\theta + \phi)$$

However, when the sum of the three terms is found, it is clear that the result is the same as the original expression.

```
> EXPRc := TERM1c + TERM2c + TERM3c;
```

$$EXPRc := \cos(2\,\theta + \phi)$$

```
>
```

Try It! (p. 131)

Repeat the previous example when x is assumed to be positive. Find conditions on x that allow all three logarithm terms to be combined into a single logarithm.

Solution

```
> restart;
```

The expression of interest is

```
> EXPR := ln((x/(x^2-1))^(2*x+2)) + (x+1)*exp(x+2);
```

$$EXPR := \ln\left(\left(\frac{x}{x^2-1}\right)^{(2x+2)}\right) + (x+1)\,\mathbf{e}^{(x+2)}$$

>

The assumption that x is positive (and, hence, real) is made with the command

```
> assume( x>0 );
> about( x );
Originally x, renamed x~:
  is assumed to be: RealRange(Open(0),infinity)
>
```

The extra information about x allows for additional simplifications of the expression.

```
> simplify( EXPR );
```

$$2\,x\sim\ln(x\sim) + 2\ln(x\sim) - \ln((x\sim-1)^2) - 2\ln(x\sim+1) - x\sim\ln((x\sim-1)^2) - 2\,x\sim\ln(x\sim+1) + \mathbf{e}^{(x\sim+2)}\,x\sim + \mathbf{e}^{(x\sim+2)}$$

>

The factorization and expanded forms are, however, unchanged

```
> EXPRf1 := factor( EXPR );
```

$$EXPRf1 := \ln\left(\left(\frac{x\sim}{(x\sim-1)\,(x\sim+1)}\right)^{(2\,x\sim+2)}\right) + \mathbf{e}^{(x\sim+2)}\,x\sim + \mathbf{e}^{(x\sim+2)}$$

```
> EXPRe1 := expand( EXPRf1 );
```

$EXPRe1 :=$

$$2\,x\sim\ln(x\sim) - 2\,x\sim\ln(x\sim-1) - 2\,x\sim\ln(x\sim+1) + 2\ln(x\sim) - 2\ln(x\sim-1) - 2\ln(x\sim+1) + \mathbf{e}^{x\sim}\,\mathbf{e}^2\,x\sim + \mathbf{e}^{x\sim}\,\mathbf{e}^2$$

```
> EXPRf2 := factor( EXPRe1 );
```

$$EXPRf2 := (x\sim+1)\,(-2\ln(x\sim-1) - 2\ln(x\sim+1) + 2\ln(x\sim) + \mathbf{e}^{x\sim}\,\mathbf{e}^2)$$

>

The fact that x>0 enables Maple to combine two of the logarithmic terms

```
> EXPRc1 := combine( EXPRf2 );
```

$$EXPRc1 := (x\sim+1)\left(-2\ln(x\sim-1) + \mathbf{e}^{(x\sim+2)} + \ln\left(\frac{x\sim^2}{(x\sim+1)^2}\right)\right)$$

>

As before, the results of expand and factor are unaffected by the assumption.

```
> EXPRe2 := expand( EXPRf1, x+2 );
```

$EXPRe2 :=$

$$2\,x\sim\ln(x\sim) - 2\,x\sim\ln(x\sim-1) - 2\,x\sim\ln(x\sim+1) + 2\ln(x\sim) - 2\ln(x\sim-1) - 2\ln(x\sim+1) + \mathbf{e}^{(x\sim+2)}\,x\sim + \mathbf{e}^{(x\sim+2)}$$

```
> EXPRc2 := factor( EXPRe2 );
```

$$EXPRc2 := (x\sim+1)\,(-2\ln(x\sim-1) - 2\ln(x\sim+1) + 2\ln(x\sim) + \mathbf{e}^{(x\sim+2)})$$

>

For the final part of this exercise, let’s begin by removing the assumption on x .

```
> x := 'x':
> about( x );
x:
  nothing known about this object
>
```

The phrase "all three logarithm terms" is referring to the logarithmic terms in EXPR2c, the last expression in Example 5-10. This expression can be extracted by cutting-and-pasting directly from EXPRc2.

```
> EXPRln := -2*ln(x)+2*ln(x-1)+2*ln(x+1);
```

$$EXPRln := -2\ln(x) + 2\ln(x-1) + 2\ln(x+1)$$

Note: alternate definition of EXPRln

Here is a way in which these terms can be isolated using Maple commands.

```
> #select( has, EXPRe2, ln );
> #EXPRln := factor( " )/(x+1);
```

While select is almost self-explanatory, you should consult the on-line help for a full description.

>

Recall that Maple will not combine the logarithmic terms because the "standard" properties are not valid when the arguments are negative and/or complex.

```
> combine( EXPRln );
```

$$-2\ln(x)+2\ln(x-1)+2\ln(x+1)$$

```
>
```

The assumption that x>0 allows for the combination of two terms (since $x+1$>0).

```
> assume( x>0 );
> combine( EXPRln );
```

$$2\ln(x\sim-1)+\ln\left(\frac{(x\sim+1)^2}{x\sim^2}\right)$$

```
>
```

To combine all three terms requires the assumption that x>1.

```
> assume( x>1 );
> combine( EXPRln );
```

$$\ln\left(\frac{(x\sim-1)^2\,(x\sim+1)^2}{x\sim^2}\right)$$

Note: a less appealing solution

An alternate, and less informative, solution to this problem is to instruct Maple to apply the combination rules for logarithms without regard for their domain of application, i.e., using only pattern matching. This is done as follows (see ?combine[ln] for details)

```
> x := 'x';
```

$$x := x$$

```
> combine( EXPRln, ln, symbolic );
```

$$\ln\left(\frac{(x-1)^2\,(x+1)^2}{x^2}\right)$$

```
>
```

5.5 Using Types and Type Conversion

Try It! (p. 133)

The results in Example 5-12 show that three of the four possible combinations of true and false can be obtained when type and hastype are applied to the same arguments. Is it possible for type to return true and hastype to return false for the same arguments?

Solution

No. Since the full expression is a subexpression of itself (a set is a subset of itself) whenever hastype returns false, type will also return false.

```
>
```

Try It! (p. 135)

Write a single Maple command that uses nested seq commands that carry out the 15-type tests for each of the 4 expressions. Be sure the results are well organized and easy to read.

Solution

```
> restart;
```

The four expressions considered in Example 5-13 (p. 135) can be assembled in a list

```
> exprs := [ exp(3), 3*Pi/2, cos(Pi/2), ln(-Pi) ];
```

$$exprs := \left[e^3, \frac{3}{2}\pi, 0, \ln(-\pi)\right]$$

Here is the list of 15 types for numeric expressions

```
> numtypes := [ numeric, positive, negative, nonneg,
>               integer, posint, negint, nonnegint, even, odd,
>               float, rational, fraction, constant, realcons ]:
>
```

The 15 type tests for each of the 4 expressions can be obtained in one command as follows:

```
> seq( print( E, [ seq( type( E, T ), T=numtypes ) ] ), E=exprs );
```

$$e^3, [false, false, false, false, false, false, false, false, false, false, false, false, false, true, true]$$

$$\frac{3}{2}\pi, [false, false, false, false, false, false, false, false, false, false, false, false, false, true, true]$$

$$0, [true, false, false, true, true, false, false, true, true, false, false, true, false, true, true]$$

$$\ln(-\pi), [false, false, false, false, false, false, false, false, false, false, false, false, false, true, false]$$

```
>
```

What If? (p. 144)

Suppose a wastewater treatment plant at a certain location along the stream. After the treatment water is mixed with the upstream water, we find that the water temperature just downstream (after mixing) is $T=26$ °C, the$DO=6.9$ mg/L, the $BOD_u=15.2$ mg/L, and the stream velocity is 20 km/day. The deoxygenation and reaeration rates are the same as the earlier upstream values. Plot the new DO sag curve and identify the critical time, t_{crit}, where the DO is at a minimum and find this minimum value. How far downstream from the treatment plant does this minimum occur? Are there any portions of the stream downstream from the treatment plant where fish cannot survive?

Solution

```
> restart;
```

Recall the basic definitions for this application:

```
> Deqn := DD = kd/(kr-kd)*BODu*(exp(-kd*t) - exp(-kr*t) )+ Do*exp(-kr*t);
```

$$Deqn := DD = \frac{kd\ BODu\ (e^{(-kd\ t)} - e^{(-kr\ t)})}{kr - kd} + Do\ e^{(-kr\ t)}$$

```
> DOconserv := DO + DD = DOsat;
```

$$DOconserv := DO + DD = DOsat$$

```
> DOeqn := op( solve( DOconserv, { DO } ) );
```

$$DOeqn := DO = -DD + DOsat$$

```
> DOeqn := subs( Deqn, DOeqn );
```

$$DOeqn := DO = -\frac{kd\ BODu\ (e^{(-kd\ t)} - e^{(-kr\ t)})}{kr - kd} - Do\ e^{(-kr\ t)} + DOsat$$

```
>
```

The parameter values downstream from the site of the contamination are

```
> DOvals := [kd = 0.4, kr = 2.0, DOo = 6.9, T = 26, BODu = 15.2, Do = 1.2, DOsat =
  8.1];
```

$$DOvals := [kd = .4, kr = 2.0, DOo = 6.9, T = 26, BODu = 15.2, Do = 1.2, DOsat = 8.1]$$

with DO_{sat} and D_0 computed from Table 5-1 and conservation of DO. The specific formula for the dissolved oxygen is

```
> DOeqn2 := subs( DOvals, DOeqn );
```

$$DOeqn2 := DO = -3.800000000\ e^{(-.4\ t)} + 2.600000000\ e^{(-2.0\ t)} + 8.1$$

A clear picture of the minimum is obtained by plotting the DO sag curve on a short interval:

```
> plot( rhs(DOeqn2), t=0..1, title=‘DO sag curve (downstream)‘ );
```

DO sag curve (downstream)




```
>
```

The DO level is lowest after approximately t_{crit}=0.77 days (about 18 hours); the lowest DO level is approximately $DO = 5.8$ mg/L. (More accurate approximations can be obtained by ''zooming in'' on the minimum; the analytic solution to this problem can be found using the techniques in Chapter 6.) Since the stream is moving at $v_{stream} = 20$ km/day, the most severe impact of the pollutant is felt approximately $t_{crit}\, v_{stream} = 15.4$ km (a little more than 9 miles) downstream from the spill. At that time the DO level of 5.8 mg/L is still about 50% above the lower limit of 4mg/L for supporting fishlife -- this spill should not seriously affect the stream's ecosystem.

Problems (pp. 146 -- 148)

Problem 1

How does Maple simplify $\sqrt{z^2}$ when z is complex? real? positive? negative? Explain all results. Since this expression involves only a single name, is there any difference between using the assume command and the assume= optional argument to simplify? (Be sure to look at the online help for any functions that you have not seen previously.)

Solution

```
> restart;
```

The expression to be studied is

```
> EXPR := sqrt( z^2 );
```

$$EXPR := \sqrt{z^2}$$

It is reasonable to manually enter the assumption and result for each of the four types. However, more efficient solutions are available. For example, the solution based on the ideas used to answer the Try It! (p. 135) is to define a list containing the four types

```
> TYPES := [  positive, negative, real, complex ];
```

$$TYPES := [positive, negative, real, complex]$$

and, then,

```
> seq( print( T, EXPR=simplify( EXPR, assume=T ) ), T=TYPES );
```

$$positive, \sqrt{z^2} = z$$

$$negative, \sqrt{z^2} = -z$$

$$real, \sqrt{z^2} = \mathrm{signum}(z)\, z$$

$$complex, \sqrt{z^2} = \mathrm{csgn}(z)\, z$$

The results when z is positive and when z is negative should be very familiar. The signum function returns the sign of its argument (+1 if positive and -1 if negative); thus, the case when z is real is a generalization of the first two cases. The csgn function is a version of signum for complex-valued arguments; see the online help for a full explanation.

```
>
```

When the assume command is used to supply information about z, we obtain:

```
> for T in TYPES do
>   assume( z, T );
>   about( z );
>   T, EXPR=simplify( EXPR );
> od;
Originally z, renamed z~:
  is assumed to be: RealRange(Open(0),infinity)
```

$$positive, \sqrt{z\sim^2} = z\sim$$

```
Originally z, renamed z~:
  is assumed to be: RealRange(-infinity,Open(0))
```

$$negative, \sqrt{z\sim^2} = -z\sim$$

```
Originally z, renamed z~:
  is assumed to be: real
```

$$real, \sqrt{z\sim^2} = \mathrm{signum}(z\sim)\, z\sim$$

```
Originally z, renamed z~:
  is assumed to be: complex
```

$$complex, \sqrt{z\sim^2} = \mathrm{csgn}(z\sim)\, z\sim$$

```
>
```

Note

While these results are exactly the same, do not assume that this will always be true. When assume= is used, unexpected assumptions might be made about temporary variables used in the simplification.

Problem 2

Determine conditions on z so that $\sqrt{\mathbf{e}^z} = \mathbf{e}^{\left(\frac{z}{2}\right)}$.

Hint

n equivalent form of this question is: when is $\sqrt{\mathbf{e}^z} - \mathbf{e}^{\left(\frac{z}{2}\right)} = 0$?

Solution

```
> restart;
```

Following the hint, the difference between the two terms is

```
> EXPR := sqrt(exp(z)) - exp(z/2);
```

$$EXPR := \sqrt{\mathbf{e}^z} - \mathbf{e}^{(1/2\,z)}$$

As expected, this expression cannot be simplified without some assumptions.

```
> simplify( EXPR );
```

$$\sqrt{\mathbf{e}^z} - \mathbf{e}^{(1/2\,z)}$$

```
>
```

Basically, we need to determine when $\mathbf{e}^z$>0. While this is not true for complex numbers, it is certainly true for all real numbers.

```
> simplify( EXPR, assume=real );
```

$$0$$

To conclude,

```
> TERM := op( 1, EXPR ):
> TERM = simplify( TERM, assume=real );
```

$$\sqrt{\mathbf{e}^z} = \mathbf{e}^{(1/2\,z)}$$

```
>
```

Problem 3

Symbolic simplification should not be overused. To see some of the potential pitfalls, consider the expression $((-2)^p)^{\left(\frac{1}{p}\right)}$.

(a)

Compute the value of this expression for p=-5, -4, -3, -2, -1, -2/3, -1/3, 0, 1/2, 1, 5/4, 3/2, 7/4, 2, 3, 4, 5.

(b)

What does Maple simplify this expression to when p is complex? positive? negative? even? odd?

(c)

How does Maple simplify this expression when the symbolic option is used in simplify?

(d)

For what values of p are the answers in parts (b) and (c) consistent?

Solution

```
> restart;
```

The expression du jour is

```
> EXPR := ((-2)^p)^(1/p);
```

$$EXPR := ((-2)^p)^{\left(\frac{1}{p}\right)}$$

```
>
```

(a) Observe that $\frac{1}{p}$ is not defined when $p=0$.

```
> subs( p=0, EXPR );
Error, division by zero
```

Omitting this value from the list of values, we are left with

```
> POWER := [ -5, -4, -3, -2, -1, -2/3, -1/3, 1/2, 1, 5/4, 3/2, 7/4, 2, 3, 4, 5
  ]:
```

For each value, Maple (automatically) simplifies the expression to

```
> seq( print( 'p'=p, 'EXPR'=EXPR ), p=POWER );
```

$$p = -5, EXPR = -(-1)^{4/5}\, 32^{1/5}$$

$$p = -4, EXPR = 16^{1/4}$$

$$p = -3, EXPR = -(-1)^{2/3}\, 8^{1/3}$$

$$p = -2, EXPR = \sqrt{4}$$

$$p = -1, EXPR = -2$$

$$p = \frac{-2}{3}, EXPR = \frac{1}{\left(-\frac{1}{2}(-2)^{1/3}\right)^{3/2}}$$

$$p = \frac{-1}{3}, EXPR = -2$$

$$p = \frac{1}{2}, EXPR = -2$$

$$p = 1, EXPR = -2$$

$$p = \frac{5}{4}, EXPR = (-2\,(-2)^{1/4})^{4/5}$$

$$p = \frac{3}{2}, EXPR = (-2\sqrt{-2})^{2/3}$$

$$p = \frac{7}{4}, EXPR = (-2\,(-2)^{3/4})^{4/7}$$

$$p = 2, EXPR = \sqrt{4}$$

$$p = 3, EXPR = (-8)^{1/3}$$

$$p = 4, EXPR = 16^{1/4}$$

$$p = 5, EXPR = (-32)^{1/5}$$

>

(b) Using seq, as in Problem 1, we find

```
> TYPES := [ positive, negative, even, odd, complex ]:
> seq( print( T, EXPR=simplify( EXPR, assume=T ) ), T=TYPES );
```

$$positive, \left((-2)^p\right)^{\left(\frac{1}{p}\right)} = 2\left((-1)^p\right)^{\left(\frac{1}{p}\right)}$$

$$negative, \left((-2)^p\right)^{\left(\frac{1}{p}\right)} = 2\left((-1)^p\right)^{\left(\frac{1}{p}\right)}$$

$$even, \left((-2)^p\right)^{\left(\frac{1}{p}\right)} = 2$$

$$odd, \left((-2)^p\right)^{\left(\frac{1}{p}\right)} = 2\,(-1)^{\left(\frac{1}{p}\right)}$$

$$complex, \left((-2)^p\right)^{\left(\frac{1}{p}\right)} = \left((-1)^p\, 2^p\right)^{\left(\frac{1}{p}\right)}$$

The first two results illustrate that the sign of the power is not important. Note that the expression never simplifies to -2, and simplifies to 2 when p is an even integer.

>

(c) The "symbolic" simplification of the expression should multiply the exponents and return -2.

```
> simplify( EXPR, symbolic );
```

$$-2$$

>

(d) The results in (b) and (c) never completely agree. They come closest when p is an odd integer. In this case there are p roots of -1 - one of which is -1.

>

Problem 4

Although it is a well-known fact from algebra that $x^2-2=(x-\sqrt{2})(x+\sqrt{2})$, this result is not obtained from factor(x^2-2);. The explanation for this can be seen in the fact that the factorizations returned by factor generally have integer coefficients. To include integer multiples of one or more specific nonintegers, called extensions to the field of integers, include these numbers as a set as the second argument to factor. For example, factor(x^2-2, { 2^(1/2) }); returns the expected factorization for x^2-2. Irrational numbers that appear in the polynomial are automatically included in the set of extensions (see Example 5-7).

Find appropriate sets of extensions that yield the full factorization of

(a)

$$x^2+4x-41$$

(b)

$$x^3-\frac{5x^2}{2}-5x+\frac{3}{2}$$

Solution

```
> restart;
```

(a)

```
> EXPR := x^2 + 4*x - 41;
```

$$EXPR := x^2+4x-41$$

```
> factor( EXPR );
```

$$x^2+4x-41$$

Since the basic factorization does not work, we need to determine an appropriate field extension. The discriminant in the quadratic formula is $\sqrt{b^2-4\,a\,c}=\sqrt{180}=6\sqrt{5}$

. This means $\sqrt{5}$ is an appropriate field extension.

```
> factor( EXPR, sqrt(5) );
```

$$-(x+2+3\sqrt{5})(-x-2+3\sqrt{5})$$

```
>
```

(b)

```
> EXPR := x^3 - 5/2*x^2 - 5*x + 3/2;
```

$$EXPR := x^3 - \frac{5}{2}x^2 - 5x + \frac{3}{2}$$

```
> factor( EXPR );
```

$$\frac{1}{2}(2x+3)(x^2-4x+1)$$

The factorization is only partially successful. The approach used in (a) can be applied to determine that $\sqrt{3}$ is an appropriate field extension.

```
> factor( EXPR, sqrt(3) );
```

$$-\frac{1}{2}(-x+2+\sqrt{3})(x-2+\sqrt{3})(2x+3)$$

Alternate determination of field extension

We could also ask Maple to explicitly find the roots of this polynomial.

```
> solve( EXPR=0, x );
```

$$\frac{-3}{2}, 2+\sqrt{3}, 2-\sqrt{3}$$

As before, this indicates that $\sqrt{3}$ is the field extension that is needed for this problem.

Note that this exercise illustrates an essential difference between factor and solve.

```
>
```

Problem 5

(a)

Use the factorization of $x^n - 1$ to obtain the roots of $x^n = 1$ with as much accuracy as possible for each n=1, 2, 3, 4, 5, 6, 7, 8.

(b)

Use the complexplot command, from the plots package, to plot all solutions to $x^n - 1$ for n=1, 2, 3, 4, 5, 6, 7, 8.

(c)

Compare your results in part (a) with the results obtained by using solve to find the solutions to $x^n = 1$.

Solution

```
> restart; with(plots):
```

To begin, let's see what how far we can get using factor:

```
> EXPR := x^n - 1;
```

$$EXPR := x^n - 1$$

```
> for n from 1 to 8 do
>   EXPR.n := factor( EXPR );
> od;
```

$$EXPR1 := x-1$$
$$EXPR2 := (x-1)(x+1)$$
$$EXPR3 := (x-1)(x^2+x+1)$$
$$EXPR4 := (x-1)(x+1)(x^2+1)$$
$$EXPR5 := (x-1)(x^4+x^3+x^2+x+1)$$
$$EXPR6 := (x-1)(x+1)(x^2+x+1)(x^2-x+1)$$

$$EXPR7 := (x-1)(x^6+x^5+x^4+x^3+x^2+x+1)$$

$$EXPR8 := (x-1)(x+1)(x^2+1)(x^4+1)$$

The first two expressions are completely factored. The solve command can help with the identification of appropriate field extensions.

```
>
```

When $n=1$ there is one root: $x=1$.

```
> solve( EXPR1=0, x );
```

$$1$$

```
> R1 := [ " ];
```

$$R1 := [1]$$

```
> P1 := complexplot( R1, style=POINT, axes=NONE,
>                    title=‘Solutions to x^1=1‘ ):
>
```

When $n=2$ the two roots are $x=1$ and $x=-1$.

```
> solve( EXPR2=0, x );
```

$$1, -1$$

```
> R2 := [ " ];
```

$$R2 := [1, -1]$$

```
> P2 := complexplot( R2, style=POINT, axes=NONE,
>                    title=‘Solutions to x^2=1‘ ):
>
```

When $n=3$ the discriminant of the quadratic term is $\sqrt{3}$. Since two of the roots are complex, it is also necessary to include I in the set of field extensions. Note that this information can also be obtained from the output from solve.

```
> solve( EXPR3=0, x );
```

$$1, -\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}$$

```
> factor( EXPR3, { I, sqrt(3) } );
```

$$\frac{1}{4}(2x+1+I\sqrt{3})(2x+1-I\sqrt{3})(x-1)$$

```
> R3 := [ "" ];
```

$$R3 := \left[1, -\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}\right]$$

```
> P3 := complexplot( R3, style=POINT, axes=NONE,
>                    title=‘Solutions to x^3=1‘ ):
>
```

When $n=4$ the only field extension that is needed is I.

```
> solve( EXPR4=0, x );
```

$$1, -1, I, -I$$

```
> factor( EXPR4, I );
```

$$(x-I)(x+I)(x+1)(x-1)$$

```
> R4 := [ "" ];
```

$$R4 := [1, -1, I, -I]$$

```
> P4 := complexplot( R4, style=POINT, axes=NONE,
>                    title=‘Solutions to x^4=1‘ ):
>
```

When $n=5$ the field extension is not so easy to specify. Maple will not accept the product of two square roots (as is displayed in the output from solve). Instead, either express the entire term as a single square root or specify each factor independently. While the results appear quite different, they are equivalent:

```
> solve( EXPR5=0, x );
```

$$1, \frac{1}{4}\sqrt{5}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}, -\frac{1}{4}\sqrt{5}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}, -\frac{1}{4}\sqrt{5}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}},$$

$$\frac{1}{4}\sqrt{5}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}$$

```
> factor( EXPR5, { I, sqrt(5), sqrt(10+2*sqrt(5)) } );
```

$$\frac{1}{1024}(8x+2+2\sqrt{5}+\%1-I\sqrt{5}\sqrt{10+2\sqrt{5}})(4x+1-\sqrt{5}+\%1)$$
$$(8x+2+2\sqrt{5}-\%1+I\sqrt{5}\sqrt{10+2\sqrt{5}})(4x+1-\sqrt{5}-\%1)(x-1)$$
$$\%1 := I\sqrt{10+2\sqrt{5}}$$

```
> factor( EXPR5, { I, sqrt(5), sqrt(2), sqrt(5+sqrt(5)) } );
```

$$\frac{1}{1024}(4x+1-\sqrt{5}-\%1)(1+4x-\sqrt{5}+\%1)(2+8x-\%1+2\sqrt{5}+I\sqrt{5}\sqrt{5+\sqrt{5}}\sqrt{2})$$
$$(8x+2+2\sqrt{5}+\%1-I\sqrt{5}\sqrt{5+\sqrt{5}}\sqrt{2})(x-1)$$
$$\%1 := I\sqrt{2}\sqrt{5+\sqrt{5}}$$

```
> R5 := [ """ ];
```

$$R5 := \left[1, \frac{1}{4}\sqrt{5}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}, -\frac{1}{4}\sqrt{5}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}, -\frac{1}{4}\sqrt{5}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}, \frac{1}{4}\sqrt{5}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right]$$

```
> P5 := complexplot( R5, style=POINT, axes=NONE,
>                      title=‘Solutions to x^5=1‘ ):
>
```

When $n=6$ the situation is much simpler -- a field extension is easily identified from the output from solve:

```
> solve( EXPR6=0, x );
```

$$1, -1, -\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}, \frac{1}{2}-\frac{1}{2}I\sqrt{3}, \frac{1}{2}+\frac{1}{2}I\sqrt{3}$$

```
> factor( EXPR6, { I, sqrt(3) } );
```

$$\frac{1}{16}(2x+1+I\sqrt{3})(2x-1-I\sqrt{3})(2x-1+I\sqrt{3})(2x+1-I\sqrt{3})(x+1)(x-1)$$

```
> R6 := [ "" ];
```

$$R6 := \left[1, -1, -\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}, \frac{1}{2}-\frac{1}{2}I\sqrt{3}, \frac{1}{2}+\frac{1}{2}I\sqrt{3}\right]$$

```
> P6 := complexplot( R6, style=POINT, axes=NONE,
>                      title=‘Solutions to x^6=1‘ ):
>
```

When $n=7$ the output from solve contains only complex trigonometric expressions. Unfortuately, these cannot be used as field extensions in factor; evalf can be used to obtain approximate numerical values for these roots. It is also possible to obtain an approximate factorization by specifying complex as the second argument to factor.

```
> solve( EXPR7=0, x );
```

$$1, \cos\left(\frac{2}{7}\pi\right)+I\sin\left(\frac{2}{7}\pi\right), -\cos\left(\frac{3}{7}\pi\right)+I\sin\left(\frac{3}{7}\pi\right), -\cos\left(\frac{1}{7}\pi\right)+I\sin\left(\frac{1}{7}\pi\right), -\cos\left(\frac{1}{7}\pi\right)-I\sin\left(\frac{1}{7}\pi\right),$$
$$-\cos\left(\frac{3}{7}\pi\right)-I\sin\left(\frac{3}{7}\pi\right), \cos\left(\frac{2}{7}\pi\right)-I\sin\left(\frac{2}{7}\pi\right)$$

```
> evalf( " );
```

$$1., .6234898018+.7818314825\,I, -.2225209335+.9749279123\,I, -.9009688678+.4338837393\,I,$$
$$-.9009688678-.4338837393\,I, -.2225209335-.9749279123\,I, .6234898018-.7818314825\,I$$

```
> factor( EXPR7, complex );
```

$$(x+.9009688679+.4338837391\,I)(x+.9009688679-.4338837391\,I)(x+.2225209340+.9749279122\,I)$$
$$(x+.2225209340-.9749279122\,I)(x-.6234898019+.7818314825\,I)(x-.6234898019-.7818314825\,I)$$

$$(x-1.)$$

```
> R7 := [ "" ];
```

$$R7 := [1., .6234898018 + .7818314825\,I, -.2225209335 + .9749279123\,I, -.9009688678 + .4338837393\,I, -.9009688678 - .4338837393\,I, -.2225209335 - .9749279123\,I, .6234898018 - .7818314825\,I]$$

```
> P7 := complexplot( R7, style=POINT, axes=NONE,
>                    title=‘Solutions to x^7=1‘ ):
>
```

When $n=8$, things are simpler:

```
> solve( EXPR8=0, x );
```

$$1, -1, I, -I, \frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2}-\frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2}-\frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}$$

```
> factor( EXPR8, { I, sqrt(2) } );
```

$$\frac{1}{16}(2x+\sqrt{2}+I\sqrt{2})(2x+\sqrt{2}-I\sqrt{2})(2x-\sqrt{2}-I\sqrt{2})(x+I)(x-I)(x+1)(x-1)(2x-\sqrt{2}+I\sqrt{2})$$

```
> R8 := [ "" ];
```

$$R8 := \left[1, -1, I, -I, \frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2}-\frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2}-\frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}\right]$$

```
> P8 := complexplot( R8, style=POINT, axes=NONE,
>                    title=‘Solutions to x^8=1‘ ):
>
```

To conclude, let’s display the plots in a 2x4 array.

```
> display( array(1..2,1..4,[[P1,P2,P3,P4],[P5,P6,P7,P8]] ) );
```




```
>
```

An animation view of these roots is another way of viewing the plots of the roots.

```
> display( [ seq( P.i, i=1..8 ) ], insequence=true );
```

Note

The animated display is omitted from the hardcopy of the Instructor’s Guide.

```
>
```

Problem 6

Find all values of the parameter a for which the functions $f(x)=x^2+a\,x+26$ and $g(x)=x^4+6\,x^3-17\,x^2-78\,x-56$ have at least one common root.

Solution

```
> restart;
```

The two functions can be defined (as expressions) as follows:

```
> f := x^2+a*x+26;
```

$$f := x^2 + a\,x + 26$$

```
> g := x^4+6*x^3-17*x^2-78*x-56;
```

$$g := x^4 + 6\,x^3 - 17\,x^2 - 78\,x - 56$$

The roots of $\mathrm{f}(x)$ and of $\mathrm{g}(x)$ are

```
> ROOTf := [ solve( f=0, x ) ];
```

$$ROOTf := \left[-\frac{1}{2}a + \frac{1}{2}\sqrt{a^2 - 104},\, -\frac{1}{2}a - \frac{1}{2}\sqrt{a^2 - 104}\right]$$

```
> ROOTg := [ solve( g=0, x ) ];
```

$$ROOTg := [4, -7, -2, -1]$$

As expected, the roots of $\mathrm{f}(x)$ depend on the parameter a.

```
>
```

To determine when the polynomials have a common root it is necessary to consider each of the eight possible pairings of roots.

```
> for rg in ROOTg do
>   for rf in ROOTf do
>     R := solve( rf=rg, { a } );
>     if R<>NULL then print( rf, rg, R ) fi;
>   od;
> od;
```

$$-\frac{1}{2}a - \frac{1}{2}\sqrt{a^2 - 104},\, 4,\, \{a = \frac{-21}{2}\}$$

$$-\frac{1}{2}a - \frac{1}{2}\sqrt{a^2 - 104},\, -7,\, \{a = \frac{75}{7}\}$$

$$-\frac{1}{2}a + \frac{1}{2}\sqrt{a^2 - 104},\, -2,\, \{a = 15\}$$

$$-\frac{1}{2}a + \frac{1}{2}\sqrt{a^2 - 104},\, -1,\, \{a = 27\}$$

```
>
```

Only four of the pairings produce a solution. The corresponding list of parameter values is

```
> A := [ -21/2, 75/7, 15, 27 ];
```

$$A := \left[\frac{-21}{2}, \frac{75}{7}, 15, 27\right]$$

To check that these values do work, look at the factorization of $\mathrm{f}(x)$ for these values of a.

```
> for a in A do
>   factor( f );
> od;
```

$$\frac{1}{2}(x - 4)(2\,x - 13)$$

$$\frac{1}{7}(x + 7)(7\,x + 26)$$

$$(x + 13)(x + 2)$$

$$(x + 26)(x + 1)$$

Good! Each of these functions does share a factor with $\mathrm{g}(x)$.

```
>
```

Problem 7

The expression EXPRe1 in Example 5-10 is not a valid simplification of EXPR for all real and complex values of x. Find values of x that give different values when inserted into EXPR and EXPRe1. Find the general conditions on x that guarantee that the two expressions are equivalent.

Solution

```
> restart;
```

The definitions of EXPR and EXPRe1 are copied from Example 5-10 (pp. 129 -- 130).

```
> EXPR := ln((x/(x^2-1))^(2*x+2)) + (x+1)*exp(x+2);
```

$$EXPR := \ln\left(\left(\frac{x}{x^2-1}\right)^{(2\,x+2)}\right) + (x+1)\,\mathbf{e}^{(x+2)}$$

```
> EXPRf1 := factor( EXPR );
```

$$EXPRf1 := \ln\left(\left(\frac{x}{(x-1)\,(x+1)}\right)^{(2\,x+2)}\right) + \mathbf{e}^{(x+2)}\,x + \mathbf{e}^{(x+2)}$$

```
> EXPRe1 := expand( EXPRf1 );
```

$$EXPRe1 := 2\,x\ln(x) - 2\,x\ln(x-1) - 2\,x\ln(x+1) + 2\ln(x) - 2\ln(x-1) - 2\ln(x+1) + \mathbf{e}^{x}\,\mathbf{e}^{2}\,x + \mathbf{e}^{x}\,\mathbf{e}^{2}$$

```
>
```

Thinking about this problem, and the information learned from the Try It! (p. 131), we expect that there might be "problems" when the argument to one or more of the logarithms in EXPRe1 is negative. For example,

```
> evalf( subs( x=1/2, [ EXPR, EXPRf1, EXPRe1 ] ) );
```

$$[17.05734562 + 3.141592654\,I,\ 17.05734562 + 3.141592654\,I,\ 17.05734562 - 9.424777962\,I]$$

Here are some more examples:

```
> for x in [ -2, -3/2, -1/2, 1/2, 3/2, I ] do
>   x, evalf( [ EXPR, EXPRe1 ] );
> od;
```

$$-2,\ [-.1890697838,\ -.1890697827 + 6.283185308\,I]$$

$$\frac{-3}{2},\ [-1.006682192 + 3.141592654\,I,\ -1.006682192 + 3.141592654\,I]$$

$$\frac{-1}{2},\ [1.835379427,\ 1.835379427]$$

$$\frac{1}{2},\ [17.05734562 + 3.141592654\,I,\ 17.05734562 - 9.424777962\,I]$$

$$\frac{3}{2},\ [83.70023768,\ 83.70023768]$$

$$I,\ [-.470053971 + 11.96529865\,I,\ -.470053972 + 5.682113346\,I]$$

Note that, in each case, the real parts are equal but that sometimes the imaginary parts differ (by a multiple of π). In general, the two expressions are equivalent for all x>1.

```
> x := 'x':
> plot( [ EXPR, EXPRe1 ], x=1..3, style=[LINE,POINT],
>       title=`Problem 7 (Chapter 5)` );
```

Problem 7 (Chapter 5)



\>

Problem 8

The polynomial with roots -1, 2, $\frac{3}{2}$, and $\sqrt{5}$ was found in Example 5-7. Find, also in expanded form, the polynomial with the same roots but with coefficients that sum to 1.

Solution

```
> restart;
```

Begin by recalling the definition of the polynomial in Example 5-7 (p. 127).

```
> POLY := (x+1)*(x-2)*(x-3/2)*(x-sqrt(5));
```

$$POLY := (x+1)\,(x-2)\left(x-\frac{3}{2}\right)(x-\sqrt{5})$$

```
> POLYe := expand( POLY );
```

$$POLYe := x^4 - x^3\sqrt{5} - \frac{5}{2}x^3 + \frac{5}{2}x^2\sqrt{5} - \frac{1}{2}x^2 + \frac{1}{2}x\sqrt{5} + 3\,x - 3\sqrt{5}$$

\>

The coefficients of the polynomial can be obtained using the coeffs command

```
> COEFFS := [ coeffs( POLYe, x ) ];
```

$$COEFFS := \left[\frac{1}{2}\sqrt{5}+3, -3\sqrt{5}, -\frac{1}{2}+\frac{5}{2}\sqrt{5}, -\sqrt{5}-\frac{5}{2}, 1\right]$$

and the sum of the coefficients can be found be adding the elements of this list

```
> SUM := add( i, i=COEFFS );
```

$$SUM := 1-\sqrt{5}$$

or by converting the list into a "sum" (i.e., type '+')

```
> convert( COEFFS, '+' );
```

$$1-\sqrt{5}$$

Note: add vs. sum

The add and sum commands appear similar, but are intended for quite different purposes. While add is designed for adding elements of a list, sum (and its inert form, Sum) is intended for use with definite and indefinite sums (including infinite series).

\>

The final polynomial is

```
> POLY1 := expand( POLYe/SUM );
```

$$POLY1 := \frac{x^4}{1-\sqrt{5}} - \frac{x^3\sqrt{5}}{1-\sqrt{5}} - \frac{5}{2}\frac{x^3}{1-\sqrt{5}} + \frac{5}{2}\frac{x^2\sqrt{5}}{1-\sqrt{5}} - \frac{1}{2}\frac{x^2}{1-\sqrt{5}} + \frac{1}{2}\frac{x\sqrt{5}}{1-\sqrt{5}} + 3\frac{x}{1-\sqrt{5}} - 3\frac{\sqrt{5}}{1-\sqrt{5}}$$

\>

To check this result observe that

```
> factor( POLY1 );
```

$$-\frac{1}{8}(1+\sqrt{5})(2x-3)(x-\sqrt{5})(x-2)(x+1)$$

```
> normal( convert( [ coeffs( POLY1, x ) ], '+' ) );
```

$$1$$

```
>
```

Problem 9

Example 5-13 presents a number of questions that are worth pursuing. Foremost is the question about the logarithm of a negative number. One way to get more insight into this question is to look at a floating-point approximation to $\ln(-\pi)$. While this can be done using evalf, find a way to achieve the same result using convert.

Solution

```
> restart;
> EXPR := ln( -Pi );
```

$$EXPR := \ln(-\pi)$$

From the list of numeric types encountered in Example 5-13 (p. 134), it seems reasonable to convert EXPR to type float.

```
> EXPR := convert( EXPR, float );
```

$$EXPR := 1.144729886 + 3.141592654\,I$$

```
>
```

Problem 10

The values tested in Example 5-13 matched different combinations of the 15 types related to numeric objects. Is it possible to find one number that matches all 15 types? If not, what is the highest number of matches that can be made with a single number?

Solution

```
> restart;
> numtypes := [ numeric, positive, negative, nonneg,
>               integer, posint, negint, nonnegint, even, odd,
>               float, rational, fraction, constant, realcons ]:
```

Since no number can be both positive and negative (or both even and odd) it's not possible to match all 15 types with a single number.

If the value is negative, you lose positive, nonnegative, posint, and nonnegint; if the value is not an integer, you could gain float and rational, but would lose integer, posint, and nonnegint.

The largest number of matches is 10 -- for any positive integer.

```
> seq( type( 1, T ), T=numtypes );
```

true, true, false, true, true, true, false, true, false, true, false, true, false, true, true

```
> NUMtrue := nops( select( has, ["], true ) );
```

$$NUMtrue := 10$$

```
>
```

Problem 11

(a)

It is well known that the sine of all integer multiples of π is zero: $\sin(n\pi)=0$ for all integers n. Add assumptions to the name n so that Maple automatically simplifies $\sin(n\pi)$ to zero. What is the value of $\cos(n\pi)$ for any integer n?

(b)

Consider the expression $\sin\left(\frac{n\pi}{2}\right)\cos\left(\frac{n\pi}{2}\right)$. Use the combine command to simplify this expression. Now, add the assumption that n is an integer. How is this extra information reflected in the original and combined expressions? (Explain any differences in the results.)

(c)

Another lesson related to assume is that assumptions should be imposed only after all other simplifications have been completed. For example, compare the results of applying combine to $\sin\left(\frac{n\pi}{2}\right)\cos\left(\frac{n\pi}{2}\right)$ with and without the assumption that n is an integer.

Solution

(a)

```
> restart;
> EXPR := [ sin( n*Pi ), cos( n*Pi ) ];
```

$$EXPR := [\sin(n\pi), \cos(n\pi)]$$

By default, no simplifications can be made to either of these expressions. However, when n is an integer:

```
> assume( n, integer );
> about( n );
Originally n, renamed n~:
  is assumed to be: integer
> EXPR;
```

$$[0, (-1)^{n\sim}]$$

This is exactly what we expect.

```
>
```

(b) and (c)

```
> restart;
> EXPR := sin(n*Pi/2) * cos(n*Pi/2);
```

$$EXPR := \sin\left(\frac{1}{2}n\pi\right)\cos\left(\frac{1}{2}n\pi\right)$$

```
> EXPRc := combine( EXPR );
```

$$EXPRc := \frac{1}{2}\sin(n\pi)$$

```
> assume( n, integer );
> EXPRc;
```

$$0$$

Now, with the assumption still in place, attempt to repeat the combine step

```
> EXPR;
```

$$\sin\left(\frac{1}{2}n\sim\pi\right)\cos\left(\frac{1}{2}n\sim\pi\right)$$

```
> EXPRc2 := combine( EXPR );
```

$$EXPRc2 := \sin\left(\frac{1}{2}n\sim\pi\right)\cos\left(\frac{1}{2}n\sim\pi\right)$$

This result suggests that some of the transformations used in combine are disabled when assumptions are used.

```
>
```

Problem 12

Properties and types are closely related. One difference is that some properties can be specified in a convenient mathematical form: e.g., assume(z>0);. Just as the type command is used to test types, the is command is used to test if a Maple object has a specific property (see the assume help worksheet). The value returned by is will be true (if the property follows from the previous assumptions), false (if the property is not always consistent with the assumptions), or, FAIL (if Maple was not able to determine whether the property is true or false).

(a)

Verify that, when Maple knows $x>2$, is(x^2+2*x+3 > 2); returns the value false and is(x^2+2*x+3 >= 2); returns the value true.

(b)

Determine appropriate properties to impose on z so that is(ln(z^2+1) > 0);

returns the value true. How can the assumptions on z be relaxed so that is(ln(z^2+1) >= 0); evaluates as true?

Solution

(a)

Correction

Note that the problem, as stated, is not correct. The correct problem should say "when Maple knows x>=-2".

```
> restart;
```

While it is not requested, let's see the results of these commands without any assumptions.

```
> is( x^2+2*x+3 > 2 );
```

$$FAIL$$

```
> is( x^2+2*x+3 >= 2 );
```

$$FAIL$$

```
>
```

Adding the (corrected) assumption, the results are as described.

```
> assume( x >= -2 );
> about( x );
Originally x, renamed x~:
  is assumed to be: RealRange(-2,infinity)
> is( x^2+2*x+3 > 2 );
```

$$false$$

```
> is( x^2+2*x+3 >= 2 );
```

$$true$$

```
>
```

(b)

```
> restart;
> EXPR := ln( z^2+1 );
```

$$EXPR := \ln(z^2+1)$$

Without assumptions, the sign of EXPR cannot be determined.

```
> is( EXPR>0 );
```

$$FAIL$$

Logarithms are positive when the argument exceeds 1. This suggests the assumption z>0.

```
> assume( z, positive );
> about( z );
Originally z, renamed z~:
  is assumed to be: RealRange(Open(0),infinity)
> is( EXPR>0 );
```

$$true$$

To check that this is the optimal assumption, note that:

```
> assume( z, nonneg );
> about( z );
Originally z, renamed z~:
  is assumed to be: RealRange(0,infinity)
> is( EXPR>0 );
```

$$false$$

Note: alternate syntax for is

An equivalent form of this command is:

```
> is( EXPR, positive );
```

$$false$$

```
>
```

```
>
```

The current assumption (z>=0) is precisely the situation in which $\ln(z^2+1)$ >= 0:

```
> is( EXPR, nonneg );
```

$$true$$

```
>
```

Problem 13

(a)

Determine at what time 99.5% of BOD_u is attained in Figure 5-1.

(b)

Find the exact time at when the BOD reaches 99.5% of the ultimate BOD for general values of the reaction rate, k_d, and the ultimate BOD, BOD_u. Explain how this time depends on both parameters.

(c)

Repeat b) for any threshold (not just 99.5% of BOD_u). That is, determine the time until a sample with reaction rate k_d reaches p% of the ultimate BOD.

Solution

```
> restart;
```

(a) This problem is a little vague. Assuming we are interested in the times when 99.5% of the available BOD has been consumed, we need an expression for the amount of oxygen consumed through time t (in days) Based on the discussion on p. 137, this would be

```
> EQN := BODu - BODu*exp(-kd*t);
```

$$EQN := BODu - BODu\, \mathbf{e}^{(-kd\, t)}$$

The parameter values used to create Figure 5-1 (p. 138) are

```
> PARAM := [ BODu=323, kd=0.228 ];
```

$$PARAM := [BODu = 323, kd = .228]$$

The time (in days) when 99.5% of the available oxygen has been consumed is

```
> solve( subs( PARAM, EQN = 0.995*BODu ), { t } );
```

$$\{t = 23.23823406\}$$

This is consistent with the graph in Figure 5-1.

```
>
```

(b) The analytic, i.e., no floating-point numbers, solution to this problem can be obtained if the floating-point constant 0.995 is replaced with the fraction 995/1000.

```
> solve( EQN = 995/1000*BODu, { t } );
```

$$\{t = \frac{\ln(200)}{kd}\}$$

Thus, the time is inversely proportional to the reaction rate (which is consistent with the units!) and completely independent of the ultimate BOD. While this might seem surprising at first, it is a very common phenomenon in all applications of exponential decay.

```
>
```

(c) The time when p% of the available BOD has been consumed is

```
> solve( EQN = p/100*BODu, { t } );
```

$$\left\{t = -\frac{\ln\left(1 - \frac{1}{100}p\right)}{kd}\right\}$$

```
>
```

Problem 14

(a)

Find, and plot, the linear function that best fits (in the least squares sense) the DO vs. temperature data in Table 5-1.

(b)

Find, and plot, the exponential function that best fits this data.

(c)

How do the two fits compare? Which looks to be the better fit? For each fit, compute the sum of the squares of the difference between the absolute error between the measured and predicted values. What does this say about the quality of the two fits?

Solution

```
> restart; with(plots): with(stats):
```

Begin by collecting the data fromTable 5-1 (p. 136):

```
> SATdata := [[0, 14.6], [1., 14.2], [2., 13.9], [3., 13.5], [4., 13.1], [5.,
  12.8], [6., 12.5], [7., 12.1], [8., 11.8], [9., 11.6], [10., 11.3], [11.,
  11.], [12., 10.8], [13., 10.5], [14., 10.3], [15., 10.1], [16., 9.9], [17.,
  9.7], [18., 9.5], [19., 9.3], [20., 9.1], [21., 8.9], [22., 8.7], [23., 8.6],
  [24., 8.4], [25., 8.3], [26., 8.1], [27., 8.], [28., 7.8], [29., 7.7], [30.,
  7.6]]:
>
```

(a) Recall that the least-squares fit to a set of data is obtained using the fit command from the stats package. Before calling this command, the data needs to be separated into two separate lists: one for the temperatures and one for the DO readings. One way to convert the data to this form is:

```
> Tdata := [ seq( DATA[1], DATA=SATdata ) ];
> DOdata := [ seq( DATA[2], DATA=SATdata ) ];
```

Tdata := [0, 1., 2., 3., 4., 5., 6., 7., 8., 9., 10., 11., 12., 13., 14., 15., 16., 17., 18., 19., 20., 21., 22., 23., 24., 25., 26., 27., 28., 29., 30.]

DOdata := [14.6, 14.2, 13.9, 13.5, 13.1, 12.8, 12.5, 12.1, 11.8, 11.6, 11.3, 11., 10.8, 10.5, 10.3, 10.1, 9.9, 9.7, 9.5, 9.3, 9.1, 8.9, 8.7, 8.6, 8.4, 8.3, 8.1, 8., 7.8, 7.7, 7.6]

The best linear fit is

```
> DOfit := fit[leastsquare[ [T,DO], DO=a*T+b, {a,b} ]]
>             ([ Tdata, DOdata ]);
```

$$DOfit := DO = -.2289516129\, T + 13.87620968$$

To see how good this fit is, plot the data and the best linear fit to the data

```
> Pdata := plot( SATdata, style=POINT, view=0..15 ):
> Pfit := plot( rhs(DOfit), T=0..30, color=BLUE ):
> display( [Pdata, Pfit], labels=[temperature,DO],
>          title=‘DO[sat] vs. temperature: data & linear fit‘ );
```

DO[sat] vs. temperature: data & linear fit




```
>
```

(b) It would be nice if the exponential fit could be found as easily as the linear fit, but ...

```
> DOfit2 := fit[leastsquare[ [T,DO], DO=a*exp(b*T), {a,b} ]]
>             ([ Tdata, DOdata ]);
```

$DOfit2 := fit_{leastsquare_{[T, DO], DO = a\, e^{(b\, T)}, \{a, b\}}}$([[0, 1., 2., 3., 4., 5., 6., 7., 8., 9., 10., 11., 12., 13., 14., 15., 16., 17., 18., 19., 20., 21., 22., 23., 24., 25., 26., 27., 28., 29., 30.], [14.6, 14.2, 13.9, 13.5, 13.1, 12.8, 12.5, 12.1, 11.8, 11.6, 11.3

11., 10.8, 10.5, 10.3, 10.1, 9.9, 9.7, 9.5, 9.3, 9.1, 8.9, 8.7, 8.6, 8.4, 8.3, 8.1, 8., 7.8, 7.7, 7.6]])

```
>
```

The problem is that, in this form, the least squares problem is nonlinear (in the parameters a and b). The standard approach to this problem is to apply a transformation to the data that makes the problem linear in the parameters. In this case this is accomplished by looking for a linear fit to the (natural) logarithm of the DO data, i.e., $\ln(DO) = \ln(a\,\mathbf{e}^{(b\,T)})$ = ln(a)+b*T . To implement this, the logarithm of the DO data is needed:

```
> lnDOdata := map( ln, DOdata );
```

$lnDOdata$:= [2.681021529, 2.653241965, 2.631888840, 2.602689685, 2.572612230, 2.549445171, 2.525728644, 2.493205453, 2.468099531, 2.451005098, 2.424802726, 2.397895273, 2.379546134, 2.351375257, 2.332143895, 2.312535424, 2.292534757, 2.272125886, 2.251291799, 2.230014400, 2.208274414, 2.186051277, 2.163323026, 2.151762203, 2.128231706, 2.116255515, 2.091864062, 2.079441542, 2.054123734, 2.041220329, 2.028148247
]

```
> lnDOfit := fit[leastsquare[ [T,lnDO], lnDO=lna+b*T, {lna,b} ]]
>               ([ Tdata, lnDOdata ]);
```

$$lnDOfit := lnDO = 2.653976605 - .02183091391\,T$$

```
> Plndata := plot( zip((x,y)->[x,y],Tdata,lnDOdata), style=POINT ):
> Plnfit := plot( rhs(lnDOfit), T=0..30, color=GREEN ):
> display( [Plndata, Plnfit], labels=[temperature,DO],
>          title='ln(DO[sat]) vs. temperature: log data & fit' );
```




```
>
```

The corresponding exponential function that fits the data is

```
> DOfit2 := DO = expand( exp( rhs( lnDOfit ) ) );
```

$$DOfit2 := DO = 14.21043573\,\mathbf{e}^{(-.02183091391\,T)}$$

```
>
```

To see how good this fit is, plot the data and the exponential fit to the data

```
> Pfit2 := plot( rhs(DOfit2), T=0..30, color=GREEN ):
> display( [Pdata, Pfit2], labels=[temperature,DO],
>          title='DO[sat] vs. temperature: data & exponential fit' );
```



>

(c) Based on a visual inspection, the exponential fit found in (b) appears to be slightly better than the linear fit found in (a).

```
> display( [Pdata, Pfit, Pfit2], labels=[temperature,DO], title=`DO[sat] vs. temperature: data, linear & exp fits` );
```



>

To compute the sum of the squares of the errors between the data and the linear function fitting the data requires the values of the linear function for each temperature in Tdata:

```
> DOlin := [ seq( rhs(DOfit), T=Tdata ) ];
```

DOlin := [13.87620968, 13.64725807, 13.41830645, 13.18935484, 12.96040323, 12.73145162, 12.50250000, 12.27354839, 12.04459678, 11.81564516, 11.58669355, 11.35774194, 11.12879033, 10.89983871, 10.67088710, 10.44193549, 10.21298387, 9.984032261, 9.755080648, 9.526129035, 9.297177422, 9.068225809, 8.839274196, 8.610322583, 8.381370970, 8.152419357, 7.923467745, 7.694516132, 7.465564519, 7.236612906, 7.007661293]

Then the sum of the square of the error at each point is

```
> add( (DOlin[i]-DOdata[i])^2, i=1..nops(DOlin) );
```

3.256758069

>

Using the same steps for the exponential function,

```
> DOexp := [ seq( rhs(DOfit2), T=Tdata ) ]:
> add( (DOexp[i]-DOdata[i])^2, i=1..nops(DOlin) );
```

$$.7243150106$$

These results confirm that the exponential function is a better fit for the DO data.

```
>
```